
What are is a higher-order function?

A function that takes one or more functions as input.

Main motivation: expressing the common denominator between
a collection of first-order functions, thus promoting code reuse!

Learning tip: HO functions may be hard to grasp at first, but will
eventually help in “seeing the forest for the trees”.

First example: abstracting case-analysis

Recall that given f :: a → c and g :: b → c, we can define

h :: Either a b → c
h (Left x) = f x
h (Right y) = g y

In other words, can define h by case-analysis.

For example:

asInt :: Either Bool Int → Int
asInt (Left b) = if b then 1 else 0
asInt (Right n) = n
isBool :: Either Bool Int → Bool
isBool (Left b) = True
isBool (Right n) = False

3 / 42



First example: abstracting case-analysis

The Prelude defines a higher-order function that “internalizes” the
principle of case-analysis over sum types, so to speak:

either :: (a → c) → (b → c) → Either a b → c
either f g (Left x) = f x
either f g (Right y) = g y

Here is how we can redefine asInt and isBool using either (and λ):

asInt = either (\b → if b then 1 else 0) (\n → n)
isBool = either (\b → True) (\n → False)

Whereas before we could spot that the two functions were
instances of a simple common “design pattern”, now they are
literally two applications of the same (higher-order) function.

First example: abstracting case-analysis

Here again:

asInt = either (\b → if b then 1 else 0) (\n → n)
isBool = either (\b → True) (\n → False)

Observe we only partially applied either . Alternatively:

asInt v = either (\b → if b then 1 else 0) (\n → n) v
isBool v = either (\b → True) (\n → False) v

but these two versions are completely equivalent.

(They are said to be “η-equivalent”.)

First example: abstracting case-analysis

Finally, recall arrow associates to the right by default:

either :: (a → c) → ((b → c) → (Either a b → c))

The type of either looks a lot like

(A ⊃ C) ⊃ ([B ⊃ C ] ⊃ [(A ∨ B) ⊃ C ])

which you can verify is a tautology. (This is a recurring theme!)

Second example: mapping over a list

Consider the following first-order functions on lists...

Second example: mapping over a list

(Add one to every element in a list of integers.)

mapAddOne :: [ Integer ] → [ Integer ]
mapAddOne [ ] = [ ]
mapAddOne (x : xs) = (1 + x) : mapAddOne xs

Example: mapAddOne [1 . . 5] = [2, 3, 4, 5, 6]

Second example: mapping over a list

(Square every element in a list of integers.)

mapSquare :: [ Integer ] → [ Integer ]
mapSquare [ ] = [ ]
mapSquare (x : xs) = (x ∗ x) : mapSquare xs

Example: mapSquare [1 . . 5] = [1, 4, 9, 16, 25]

Second example: mapping over a list

(Compute the length of each list in a list of lists.)

mapLength :: [ [a ]] → [ Int ]
mapLength [ ] = [ ]
mapLength (x : xs) = length x : mapLength xs

Example: mapLength ["hello", "world!"] = [5, 6]‘

10 / 42



Second example: mapping over a list

GCD = “apply some transformation to every element of a list”

We can internalize this as a higher-order function:

map :: (a → b) → [a ] → [b ]
map f [ ] = [ ]
map f (x : xs) = (f x) : map f xs

For example:

mapAddOne = map (1+)
mapSquare = map (\n → n ∗ n)
mapLength = map length

Some useful functions on functions

The “currying” and “uncurrying” principles:

curry :: ((a, b) → c) → (a → b → c)
curry f x y = f (x , y)
uncurry :: (a → b → c) → ((a, b) → c)
uncurry g (x , y) = g x y

Or equivalently:

curry f = \x → \y → f (x , y)
uncurry g = \(x , y) → g x y

Example: map (uncurry (+)) [(0, 1), (2, 3), (4, 5)] = [1, 5, 9]

Logically: (A ∧ B) ⊃ C ⇐⇒ A ⊃ (B ⊃ C).

Some useful functions on functions

The principle of sequential composition:

(◦) :: (b → c) → (a → b) → (a → c)
(g ◦ f ) x = g (f x)

Example: map ((+1) ◦ (∗2)) [0 . . 4] = [1, 3, 5, 7, 9]

Logically: transitivity of implication.

Some useful functions on functions

The principle of exchange:

flip :: (a → b → c) → (b → a → c)
flip f x y = f y x

The principle of weakening:

const :: b → (a → b)
const x y = x

The principle of contraction:

dupl :: (a → a → b) → (a → b)
dupl f x = f x x

More higher-order functions on lists

The Haskell Prelude and Standard Library define a number of HO
functions that capture common ways of manipulating lists...

More higher-order functions on lists

filter :: (a → Bool) → [a ] → [a ]
filter p [ ] = [ ]
filter p (x : xs)

| p x = x : filter p xs
| otherwise = filter p xs

Examples:

> filter (>3) [1 . . 5]
[4, 5]
> filter Data.Char .isUpper "Glasgow Haskell Compiler"
"GHC"

More higher-order functions on lists

all , any :: (a → Bool) → [a ] → Bool
all p [ ] = True
all p (x : xs) = p x && all p xs
any p [ ] = False
any p (x : xs) = p x || any p xs

Examples: all (>3) [1 . . 5] = False, any (>3) [1 . . 5] = True.

More higher-order functions on lists

takeWhile, dropWhile :: (a → Bool) → [a ] → [a ]
takeWhile p [ ] = [ ]
takeWhile p (x : xs)

| p x = x : takeWhile p xs
| otherwise = [ ]

dropWhile p [ ] = [ ]
dropWhile p (x : xs)

| p x = dropWhile p xs
| otherwise = x : xs

Examples: takeWhile (>3) [1 . . 5] = [ ],
takeWhile (<3) [1 . . 5] = [1, 2],
dropWhile (<3) [1 . . 5] = [3, 4, 5].

18 / 42



More higher-order functions on lists

concatMap :: (a → [b ]) → [a ] → [b ]
concatMap f [ ] = [ ]
concatMap f (x : xs) = f x ++ concatMap f xs

Examples:

> concatMap (\x → [x ]) [1 . . 5]
[1, 2, 3, 4, 5]
> concatMap (\x → if x ‘mod ‘ 2 ≡ 1 then [x ] else [ ]) [1 . . 5]
[1, 3, 5]
> concatMap (\x → concatMap (\y → [x . . y ]) [1 . . 3]) [1 . . 3]
[1, 1, 2, 1, 2, 3, 2, 2, 3, 3]

Note concatMap f = concat ◦ map f .

foldr : the Swiss army knife of list functions

Remarkably, all of the preceding higher-order list functions, and
many other functions besides, can be defined as instances of a
single higher-order function!

foldr : the Swiss army knife of list functions

Suppose want to write a function [a ] → b inductively over lists.

We provide a “base case” v :: b.

We provide an “inductive step” f :: a → b → b.

Putting these together, we get a recursive definition:

h :: [a ] → b
h [ ] = v
h (x :: xs) = f x (h xs)

foldr : the Swiss army knife of list functions

Since this schema is completely generic in the “base case” and the
“inductive step”, we can internalize it as a higher-order function:

foldr :: (a → b → b) → b → [a ] → b
foldr f v [ ] = v
foldr f v (x : xs) = f x (foldr f v xs)

Here are some examples:

filter p = foldr (\x xs → if p x then x : xs else xs) [ ]
all p = foldr (\x b → p x && b) True
takeWhile p = foldr (\x xs → if p x then x : xs else [ ]) [ ]
concatMap f = foldr (\x ys → f x ++ ys) [ ]

And let’s look at some more...

22 / 42



foldr : the Swiss army knife of list functions

sum :: Num a ⇒ [a ] → a
sum [ ] = 0
sum (x : xs) = x + sum xs

may be summarized as:

sum = foldr (+) 0

23 / 42



foldr : the Swiss army knife of list functions

product :: Num a ⇒ [a ] → a
product [ ] = 1
product (x : xs) = x ∗ product xs

may be summarized as:

product = foldr (∗) 1

foldr : the Swiss army knife of list functions

length :: [a ] → Int
length [ ] = 0
length (x : xs) = 1 + length xs

may be summarized as:

length = foldr (\x n → 1 + n) 0 = foldr (const (1+)) 0

foldr : the Swiss army knife of list functions

concat :: [ [a ]] → [a ]
concat [ ] = [ ]
concat (xs : xss) = xs ++ concat xss

may be summarized as:

concat = foldr (++) [ ]

foldr : the Swiss army knife of list functions

copy :: [a ] → [a ]
copy [ ] = [ ]
copy (x : xs) = x : copy xs

may be summarized as:

copy = foldr (:) [ ]

foldr : the Swiss army knife of list functions

(a somewhat more subtle example:)

(++) :: [a ] → [a ] → [a ]
[ ] ++ ys = ys
(x : xs) ++ ys = x : (xs ++ ys)

may be summarized as:

(++) = foldr (\x g → (x :) ◦ g) id

Aside: folding from the left

foldr (+) 0 [1, 2, 3, 4, 5]
= 1 + foldr (+) 0 [2, 3, 4, 5]
= 1 + (2 + foldr (+) 0 [3, 4, 5]
= 1 + (2 + (3 + foldr (+) 0 [4, 5]
= 1 + (2 + (3 + (4 + foldr (+) 0 [5]
= 1 + (2 + (3 + (4 + (5 + foldr (+) 0 [ ]))))
= 1 + (2 + (3 + (4 + (5 + 0))))
= 1 + (2 + (3 + (4 + 5)))
= 1 + (2 + (3 + 9))
= 1 + (2 + 12)
= 1 + 14
= 15

Observe that additions are performed right-to-left.

Aside: folding from the left

Sometimes we want to go left-to-right:

foldl :: (b → a → b) → b → [a ] → b
foldl f v [ ] = v
foldl f v (x : xs) = foldl f (f v x) xs

Example:

foldl (+) 0 [1, 2, 3, 4, 5]
= foldl (+) 1 [2, 3, 4, 5]
= foldl (+) 3 [3, 4, 5]
= foldl (+) 6 [4, 5]
= foldl (+) 10 [5]
= foldl (+) 15 [ ]
= 15

(Q: does this remind you of something from Lecture 1?)

Higher-order functions over trees

Recall our data type of binary trees with labelled nodes:

data BinTree a = Leaf | Node a (BinTree a) (BinTree a)
deriving (Show , Eq)

It supports a natural analogue of the map function on lists:

mapBT :: (a → b) → BinTree a → BinTree b
mapBT f Leaf = Leaf
mapBT f (Node x tL tR) = Node (f x)

(mapBT f tL) (mapBT f tR)

Higher-order functions over trees

Higher-order functions over trees

It also supports a natural analogue of foldr :

foldBT :: b → (a → b → b → b) → BinTree a → b
foldBT v f Leaf = v
foldBT v f (Node x tL tR) = f x

(foldBT v f tL) (foldBT v f tR)

For example:

nodes = foldBT 0 (\x m n → 1 + m + n)
leaves = foldBT 1 (\x m n → m + n)
height = foldBT 0 (\x m n → 1 + max m n)
mirror = foldBT Leaf (\x tL′ tR ′ → Node x tR ′ tL′)

Type classes

By now we’ve seen several examples of polymorphic functions with
type class constraints, e.g.:

sort :: Ord a ⇒ [a ] → [a ]
lookup :: Eq a ⇒ a → [(a, b)] → Maybe b
sum, product :: Num a ⇒ [a ] → a

Intuitively, these constraints express minimal requirements on the
type a needed to define these functions.

Type classes

Formally, a type class is defined by specifying the type signatures
of operations, possibly together with default implementations of
some operations in terms of others. For example:

class Eq a where
(≡), ( ̸≡) :: a → a → Bool
x ̸≡ y = not (x ≡ y)
x ≡ y = not (x ̸≡ y)

We show the constraint is satisfied by providing an instance, e.g.:

instance Eq Bool where
False ≡ b = not b
True ≡ b = b

Type classes

Sometimes need hereditary constraints to define instances:

instance Eq a ⇒ Eq [a ] where
[ ] ≡ [ ] = True
(x : xs) ≡ (y : ys) = x ≡ y && xs ≡ ys

≡ = False

Type classes

Possible for one class to inherit from another, e.g.:

class Eq a ⇒ Ord a where
compare :: a → a → Ordering
(<), (⩽), (>), (⩾) :: a → a → Bool
max , min :: a → a → a
compare x y = if x ≡ y then EQ

else if x ⩽ y then LT
else GT

x < y = case compare x y of {LT → True; → False}
x ⩽ y = case compare x y of {GT → False; → True}
x > y = case compare x y of {GT → True; → False}
x ⩾ y = case compare x y of {LT → False; → True}
max x y = if x ⩽ y then y else x
min x y = if x ⩽ y then x else y

Type classes

(That looks complicated, but basically you only need to implement
(⩽) to define an Ord instance, assuming you already have Eq.)

Type classes

Often implicit that operations should obey certain laws.

For example, (≡) should be reflexive, symmetric, and transitive.

Similarly, (⩽) should be a total ordering.

These expectations may be described in the documentation of a
type class, but are not enforced by the language.

Type classes

Type classes are a very cool feature of Haskell, but it is worth
mentioning that in a certain sense they may be seen as “just” a
convenient mechanism for defining higher-order functions.

Indeed, a constraint may always be replaced by the types of the
operations in (a minimal defn of) the corresponding type class...

40 / 42



Type classes

Replace sort :: Ord a ⇒ [a ] → [a ] by

sortHO :: (a → a → Bool) → [a ] → [a ]

Replace lookup :: Eq a ⇒ a → [(a, b)] → Maybe b by

lookupHO :: (a → a → Bool) → a → [(a, b)] → Maybe b

etc.

Type classes

Though this translation does have certain drawbacks...

Most significantly, every call to a function with constraints would
have to pass extra arguments, corresponding to the
implementation of the type class instance.

In effect, type classes are useful because these “semantically
implicit” arguments are automatically inferred by the type checker.

(Although that too has its drawbacks – for example in Haskell it is
only possible to define one instance of a type class for a given type,
although there may not be a single canonical instance.)

